

Le modèle grid.mod (0.3)

Pour TeXgraph 1.97


5 août 2014

Résumé

Ce modèle permet le tracé de droites graduées, d'axes Ox ou Oy, et de grilles, en utilisant le modèle *draw2d.mod*.

Table des matières

1	Introduction	2
2	Dessiner une droite graduée	2
3	Dessiner un axe horizontal ou un axe vertical	3
4	Dessiner un repère	5
5	Dessiner une boîte graduée	6
6	Dessiner une grille	8



1 Introduction

Le modèle *grid.mod* charge le modèle *draw2d.mod* puis les macros du fichier *grid.mac*, celles-ci permettent le dessin de droites graduées, des axes Ox ou Oy , d'un repère xOy , et de grilles. L'utilisateur a donc principalement les macros suivantes à sa disposition :

- **draw("gradline",...)** : pour dessiner une droite graduée ;
- **draw("axeX",...)** : pour un axe horizontal ;
- **draw("axeY",...)** : pour un axe vertical ;
- **draw("axes",...)** : pour dessiner un repère xOy ;
- **draw("gradBox",...)** : pour dessiner une boîte graduée ;
- **draw("grid",...)** : pour dessiner une grille.

Au chargement du modèle, les variables globales *TeXifyLabels* et *dollar* ont la valeur 1, ce qui fait que les labels correspondant aux graduations seront automatiquement entourés du symbole dollar (\$), et que lors des exports vers les formats *eps*, *pdf* et *svg*, les labels seront compilés à part par TeX, puis transformés en chemins pour être ensuite réintégrés et dessinés dans le graphique.

2 Dessiner une droite graduée

draw("gradline", <droite [A,u]>, [options])

La droite est représentée par un point A (affixe) et un vecteur directeur \mathbf{u} (affixe supposé non nul), les graduations principales correspondront aux abscisses entières dans le repère (A, \mathbf{u}) , c'est à dire aux points de la forme $A + n\mathbf{u}$ avec n entier. Pour le point d'abscisse n , le texte du label correspondant sera :

$$\frac{(\text{originnum} + \text{unit} * n) \text{ "labeltext" }}{\text{labelden}}$$

Par défaut, *originnum* vaut 0, *unit* vaut 1, "*labeltext*" est une chaîne vide, et *labelden* vaut 1. Autrement dit, par défaut, c'est la valeur de n qui est affichée.

Les options sont :

- **showaxe := { 0/1 }** indique si l'axe doit être dessiné ou non, 1 par défaut,
- **limits := { jump ou $n_1 + i * n_2$ }** définit la portion de droite à tracer, avec la valeur *jump* (valeur par défaut), c'est toute la droite qui est tracée. Avec la valeur $n_1 + i * n_2$ ce sera le segment $[A + n_1 \mathbf{u}; A + n_2 \mathbf{u}]$,
- **gradlimits := { jump ou $n_1 + i * n_2$ }** définit la portion de droite à graduer, avec la valeur *jump* (valeur par défaut), c'est toute la droite qui est graduée. Avec la valeur $n_1 + i * n_2$ ce sera le segment $[A + n_1 \mathbf{u}; A + n_2 \mathbf{u}]$,
- **unit := { valeur strictement positive }** définit l'unité pour une graduation principale, 1 par défaut,
- **nbsubdiv := { entier positif }** définit le nombre de graduations secondaires entre deux graduations principales consécutives (0 par défaut),
- **tickpos := { nombre entre 0 et 1 }** définit la position des graduations par rapport à l'axe, avec la valeur 0 elles seront entièrement au-dessus de l'axe, la valeur par défaut est 0.5,
- **tickdir := { jump ou vecteur non nul }** définit la direction des graduations, avec la valeur *jump* (valeur par défaut), c'est la direction orthogonale à l'axe,
- **xyticks := { longueur }** définit la longueur des graduations principales (cm), la valeur par défaut est 0.2. Les graduations secondaires ont une longueur correspondant à la moitié,
- **xylabelsep := { distance }** définit la distance entre les graduations principales et leurs labels,
- **originpos := { jump/center/left/right }** définit la position du label à l'origine (point A). Avec la valeur *jump* le label n'apparaîtra pas, avec la valeur *center* (valeur par défaut) il sera centré par rapport à la graduation, avec la valeur *right* il sera décalé dans le sens du vecteur \mathbf{u} de 0.25 cm, et avec la valeur *left* il sera décalé dans le sens du vecteur $-\mathbf{u}$ de 0.25 cm,
- **originnum := { nombre }** définit le numérateur du label à l'origine (point A), la valeur par défaut est 0,
- **labelpos := { jump/top/bottom }** définit la position des labels par rapport à l'axe, avec la valeur *jump* ils ne sont pas affichés, avec la valeur *top* ils sont affichés au-dessus de l'axe (orienté par le vecteur \mathbf{u}), et avec la valeur *bottom* (valeur par défaut), ils sont affichés en-dessous.

- **labelstyle** := $\langle \text{ortho/left/right/top/bottom/...} \rangle$ définit le style des labels. Avec la valeur *ortho* (valeur par défaut) ils sont orthogonaux à l'axe,
- **labelden** := $\langle \text{entier} \rangle$ définit le dénominateur des labels, 1 par défaut,
- **labeltext** := $\langle \text{"texte"} \rangle$ définit le texte ajouté au numérateur des labels, c'est une chaîne vide par défaut,
- **nbdec** := $\langle \text{entier positif} \rangle$ définit le nombre de décimales affichées, 2 par défaut,
- **numericFormat** := $\langle 0/1/2 \rangle$ définit le format d'affichage des nombres, 0 : affichage standard de TeXgraph, 1 : affichage en notation scientifique, et 2 : affichage en notation ingénieur (c'est à dire : éventuellement un signe $-$, puis un nombre de l'intervalle $[1; 1000[$ suivi de la lettre *E* et d'un exposant multiple de 3). La valeur par défaut de ce paramètre est 0,
- **legend** := $\langle \text{"texte"} \rangle$ permet d'ajouter une légende à l'axe, ce texte et une chaîne vide par défaut. Le style est définie avec la variable *LabelStyle*,
- **legendpos** := $\langle \text{nombre entre 0 et 1} \rangle$ définit la position de la légende le long de l'axe, la valeur par défaut est 0.975,
- **legendsep** := $\langle \text{distance} \rangle$ définit la distance entre la légende et son point d'ancrage, la valeur par défaut est 0.4.
- **legendangle** := $\langle \text{angle en degrés} \rangle$ définit l'angle de la légende par rapport à l'horizontale, la valeur par défaut est *jump* et dans ce cas l'angle est le même que celui des labels.



Exemple

```
\begin{texgraph}[name=gradline,file]
Include "grid.mod";
Graph image = [
Arrows :=1, CurrentArrow :=Carrow, size(8),
draw("gradline", [3.25*i,1+i/2], [limits :=-4+4*i, legend :="Axe", legendpos :=0.5]),
CurrentArrow :=CParrow,
draw("gradline", [-3,1], [legend :="demo", labeltext :="\pi", labelden :=3,
unit :=2, nbsubdiv :=1]),
CurrentArrow :=CTarrow',
draw("gradline", [3-4*i,-1.25+i/5], [legend :="A", labelstyle :=bottom, gradlimits :=-1+5*i,
nbsubdiv :=3, unit :=1.411, nbdec :=3, LabelSize :=scriptsize])
];
\end{texgraph}
```




FIGURE 1: Exemple de droites graduées

3 Dessiner un axe horizontal ou un axe vertical

draw("axeX", <[origine, pas]>, [options])

ou

draw("axeY", <[origine, pas]>, [options])

L'*origine* est un point (affixe), l'abscisse de l'*origine* sera sa partie réelle pour un axe horizontal, et sa partie imaginaire pour un axe vertical. Le *pas* un réel non nul et qui peut être négatif.

Les options sont :

- **showaxe** := $\langle 0/1 \rangle$ indique si l'axe doit être dessiné ou non, 1 par défaut,
- **limits** := $\langle \text{jump ou } x1+i*x2 \rangle$ définit la portion de droite à tracer, avec la valeur *jump* (valeur par défaut), c'est toute la droite qui est tracée. Avec la valeur $x1+i*x2$ ce sera le segment constitué des points dont l'abscisse (pour Ox) ou l'ordonnée (pour Oy) est dans l'intervalle $[x_1; x_2]$,
- **gradlimits** := $\langle \text{jump ou } x1+i*x2 \rangle$ définit la portion de droite à graduer, avec la valeur *jump* (valeur par défaut), c'est toute la droite qui est graduée. Avec la valeur $x1+i*x2$ ce sera le segment constitué des points dont l'abscisse (pour Ox) ou l'ordonnée (pour Oy) est dans l'intervalle $[x_1; x_2]$,
- **unit** := $\langle \text{valeur strictement positive} \rangle$ définit l'unité pour une graduation principale, 1 par défaut,
- **nbsubdiv** := $\langle \text{entier positif} \rangle$ définit le nombre de graduations secondaires entre deux graduations principales consécutives (0 par défaut),
- **tickpos** := $\langle \text{nombre entre 0 et 1} \rangle$ définit la position des graduations par rapport à l'axe, avec la valeur 0 elles seront entièrement au-dessus de l'axe, la valeur par défaut est 0.5,
- **tickdir** := $\langle \text{jump ou vecteur non nul} \rangle$ définit la direction des graduations, avec la valeur *jump* (valeur par défaut), c'est la direction orthogonale à l'axe,
- **xyticks** := $\langle \text{longueur} \rangle$ définit la longueur des graduations principales (cm), la valeur par défaut est 0.2. Les graduations secondaires ont une longueur correspondant à la moitié,
- **xylabelsep** := $\langle \text{distance} \rangle$ définit la distance entre les graduations principales et leurs labels, 0.1 par défaut,
- **originpos** := $\langle \text{jump/center/left/right/top/bottom} \rangle$ définit la position du label à l'origine. Avec la valeur *jump* le label n'apparaîtra pas, avec la valeur *center* (valeur par défaut) il sera centré par rapport à la graduation, pour l'axe Ox : avec la valeur *right* il sera à droite de la graduation (0.25 cm), et à gauche avec la valeur *left*. Pour l'axe Oy : avec la valeur *top* il sera au-dessus de la graduation (0.25 cm), et en-dessous avec la valeur *bottom*,
- **originnum** := $\langle \text{nombre} \rangle$ définit le numérateur du label à l'origine, la valeur par défaut est la partie réelle de l'*origine* pour l'axe Ox , t la partie imaginaire pour l'axe Oy ,
- **labelpos** := $\langle \text{jump/top/bottom/left/right} \rangle$ définit la position des labels par rapport à l'axe, avec la valeur *jump* ils ne sont pas affichés, avec la valeur *top* ils sont affichés au-dessus de l'axe pour Ox , et en-dessous avec la valeur *bottom* (valeur par défaut). Pour l'axe Oy ils sont affichés à gauche avec la valeur *left* (valeur par défaut) et à droite avec la valeur *right*,
- **labelstyle** := $\langle \text{ortho/left/right/top/bottom/...} \rangle$ définit le style des labels. Avec la valeur *ortho* ils sont orthogonaux à l'axe. Par défaut la valeur est *top* pour l'axe Ox et *right* pour l'axe Oy ,
- **labelden** := $\langle \text{entier} \rangle$ définit le dénominateur des labels, 1 par défaut,
- **labeltext** := $\langle \text{"texte"} \rangle$ définit le texte ajouté au numérateur des labels, c'est une chaîne vide par défaut,
- **nbdeci** := $\langle \text{entier positif} \rangle$ définit le nombre de décimales affichées, 2 par défaut,
- **numericFormat** := $\langle 0/1/2 \rangle$ définit le format d'affichage des nombres, 0 : affichage standard de TeXgraph, 1 : affichage en notation scientifique, et 2 : affichage en notation ingénieur (c'est à dire : éventuellement un signe —, puis un nombre de l'intervalle $[1; 1000[$ suivi de la lettre *E* et d'un exposant multiple de 3). La valeur par défaut de ce paramètre est 0,
- **legend** := $\langle \text{"texte"} \rangle$ permet d'ajouter une légende à l'axe, ce texte et une chaîne vide par défaut,
- **legendpos** := $\langle \text{nombre entre 0 et 1} \rangle$ définit la position de la légende le long de l'axe, la valeur par défaut est 0.975,
- **legendsep** := $\langle \text{distance} \rangle$ définit la distance (cm) entre la légende et son point d'ancrage, la valeur par défaut est 0.4,
- **legendangle** := $\langle \text{angle en degrés} \rangle$ définit l'angle de la légende par rapport à l'horizontale, la valeur par défaut est *jump* et dans ce cas l'angle est le même que celui des labels (0 par défaut).



Exemple

```
\begin{texgraph}[name=axeXaxeY,file]
Include "grid.mod";
Graph image = [
Arrows :=1, CurrentArrow :=CParrow', view(-5,5,-4,4), Marges(0,0,0,0), size(8),
draw("axeX", [3*i,1], [limits :=-4+4*i, legend :="$x$", ],),
draw("axeX", [2*sqrt(2)-3*i,-sqrt(2)],
[limits :=-4+4*i, legend :="$x$", labeltext :="\sqrt{2}", nbsubdiv :=1, originnum :=2]),
draw("axeY", [-4.5,1], [legend :="$y$", unit :=2]),
draw("axeY", [4.5,-1], [legend :="$y$",labelpos :=right, labelstyle :=left])
];
\end{texgraph}
```




FIGURE 2: Exemple d'axes horizontaux et verticaux

4 Dessiner un repère

draw("axes", <[origine, Xpas+i*Ypas]>, [options])

Cette macro dessine un axe horizontal et un axe vertical. L'*origine* est un point (affixe), l'abscisse de l'*origine* sera sa partie réelle et l'ordonnée sa partie imaginaire. Les réels *Xpas* et *Ypas* sont non nuls et peuvent être négatifs. Les options sont les mêmes que celles décrites précédemment à la différence près qu'elles doivent être une liste de deux valeurs (et non plus une seule valeur), la première valeur s'appliquera à l'axe *Ox* et la deuxième à l'axe *Oy*. Nous indiquons seulement la valeur par défaut des options :

- **showaxe** := < [1, 1] >
- **limits** := < [jump, jump] >
- **gradlimits** := < [jump, jump] >
- **unit** := < [1, 1] >
- **nbsubdiv** := < [0, 0] >
- **tickpos** := < [0.5, 0.5] >
- **tickdir** := < [jump, jump] >
- **xyticks** := < [0.2, 0.2] >
- **xylabelsep** := < [0.1, 0.1] >
- **originpos** := < [right, left] >
- **originnum** := < [Re(origine), Im(origine)] >
- **labelpos** := < [bottom, left] >
- **labelstyle** := < [top, right] >
- **labelden** := < [1, 1] >
- **labeltext** := < ["", ""] >
- **nbdeci** := < [2, 2] >
- **numericFormat** := < [0, 0] >

- `legend := < ["", ""] >`
- `legendpos := < [0.975, 0.975] >`
- `legendsep := < [0.4, 0.4] >`
- `legendangle := < [jump,jump] >`

Des options supplémentaires permettent d'ajouter le dessin d'une grille, cette grille occupe toute la fenêtre (les axes seront dessinés par dessus) :

- `grid := < 0/1 >`, permet d'afficher ou non une grille
- `gridwidth := < épaisseur >`, épaisseur des traits de la grille principale, valeur de *Width* par défaut,
- `subgridwidth := < épaisseur >`, épaisseur des traits de la grille secondaire, valeur de *Width/2* par défaut,
- `gridcolor := < couleur >`, couleur de la grille principale, *gray* par défaut,
- `subgridcolor := < couleur >`, couleur de la grille secondaire, *lightgray* par défaut,
- `gridstyle := < style de ligne >`, indique le style de ligne pour la grille principale, *solid* par défaut,
- `subgridstyle := < style de ligne >`, indique le style de ligne pour la grille secondaire, *solid* par défaut,



Exemple

```
\begin{texgraph}[name=axes, file]
Include "grid.mod";
Graph image = [
view(-10,10,-5,5), Marges(0.25,0.25,0.25,0.25), size(16),
Arrows :=1, CurrentArrow :=CParrow,
SaveWin(),
view(-10,-1,-4,5), ChangeWinTo([-2*pi-3*i,2*pi+3*i]),
draw("axes", [0,pi+i],[labeltext :=["\pi",""], legend :=["$x$", "$y$"]]),
RestoreWin(), SaveWin(),
view(1,10,-4,5), ChangeWinTo([-4-4*i,5+5*i]),
draw("axes", [-4-4*i,1+i], [grid :=1, legendpos :=[1,1], legend :=["$x$", "$y$"],
originpos :=[center,center],nbsubdiv :=[1,1],subgridstyle :=userdash ]),
RestoreWin()
];
\end{texgraph}
```




FIGURE 3: Exemples de repères

5 Dessiner une boîte graduée

`draw("gradBox", <[x1+i*y1, x2+i*y2, Xpas+i*Ypas]>, [options])`

Cette macro dessine une boîte représentant le pavé $[x_1; x_2] \times [y_1; y_2]$. Les réels X_{pas} et Y_{pas} sont non nuls et peuvent être négatifs. Les options sont quasiment les mêmes que celles décrites dans la section précédente sauf peut-être pour les valeurs par défaut. Nous indiquons seulement la valeur par défaut des options :

- `gradlimits` := $\langle [\text{jump}, \text{jump}] \rangle$
- `unit` := $\langle [1, 1] \rangle$
- `nbsubdiv` := $\langle [0, 0] \rangle$
- `tickpos` := $\langle [0, 1] \rangle$ (graduations tournées vers l'intérieur)
- `tickdir` := $\langle [\text{jump}, \text{jump}] \rangle$
- `xyticks` := $\langle [0.2, 0.2] \rangle$
- `xylabelsep` := $\langle [0.1, 0.1] \rangle$
- `originpos` := $\langle [\text{center}, \text{center}] \rangle$
- `originnum` := $\langle [x1, y1] \rangle$
- `labelpos` := $\langle [\text{bottom}, \text{left}] \rangle$
- `labelstyle` := $\langle [\text{top}, \text{right}] \rangle$
- `labelden` := $\langle [1, 1] \rangle$
- `labeltext` := $\langle ["", ""] \rangle$
- `nbdeci` := $\langle [2, 2] \rangle$
- `numericFormat` := $\langle [0, 0] \rangle$
- `legend` := $\langle ["", ""] \rangle$
- `legendpos` := $\langle [0.5, 0.5] \rangle$
- `legendsep` := $\langle [-0.8, -1] \rangle$
- `legendangle` := $\langle [\text{jump}, 90] \rangle$

Une option supplémentaire permet d'ajouter un titre au-dessus de la boîte :

`title` := $\langle \text{"Titre"} \rangle$, c'est une chaîne vide par défaut.

Comme pour les axes, il est possible d'ajouter le dessin d'une grille :

- `grid` := $\langle 0/1 \rangle$, permet d'afficher ou non une grille
- `gridwidth` := $\langle \text{épaisseur} \rangle$, épaisseur des traits de la grille principale, valeur de *Width* par défaut,
- `subgridwidth` := $\langle \text{épaisseur} \rangle$, épaisseur des traits de la grille secondaire, valeur de *Width*/2 par défaut,
- `gridcolor` := $\langle \text{couleur} \rangle$, couleur de la grille principale, *gray* par défaut,
- `subgridcolor` := $\langle \text{couleur} \rangle$, couleur de la grille secondaire, *lightgray* par défaut,
- `gridstyle` := $\langle \text{style de ligne} \rangle$, indique le style de ligne pour la grille principale, *solid* par défaut,
- `subgridstyle` := $\langle \text{style de ligne} \rangle$, indique le style de ligne pour la grille secondaire, *solid* par défaut,



Exemple

```
\begin{texgraph}[name=gradBox, file]
Include "grid.mod";
Graph image = [
view(-5,4,-5,5), size(12,0),
window := [-pi-4*i, pi+4*i], //boite
draw("gradBox", [window, pi/2+i], //boite et pas
[originnum := [-2,-4], labeltext := ["\pi", ""], labelden := [2,1],
grid := 1, legend := ["Abscisses", "Ordonnées"],
nbsubdiv := [1,1], title := {\large \textbf{Titre boîte}}]
),
SaveWin(), view(window),
draw("cartesian", sin(3*x)+x, [Width := 12, x := [-pi, pi], Color := red,
legend := "$C_f$", labelpos := 0.4, labeldir := South]),
draw("straightL", y-x=0, [Width := 8, Color := darkblue, legend := "$D$",
labelpos := 0.75]), //droite
RestoreWin()
];
\end{texgraph}
```




FIGURE 4: Exemple de boîte graduée

6 Dessiner une grille

`draw("grid", <[x1+i*y1, x2+i*y2]>, [options])`

Cette macro dessine une grille dans le pavé $[x_1; x_2] \times [y_1; y_2]$. Les options sont :

- `unit := < liste de deux nombres non nuls >`, définit l'unité sur Ox et sur Oy pour la graduation principale, $[1, 1]$ par défaut,
- `nbsubdiv := < liste de deux entiers positifs >`, définit le nombre de graduations secondaires entre deux graduations principales consécutives sur Ox et sur Oy , $[0, 0]$ par défaut,
- `gridwidth := < épaisseur >`, épaisseur des traits de la grille principale, 8 par défaut,
- `subgridwidth := < épaisseur >`, épaisseur des traits de la grille secondaire, 4 par défaut,
- `gridcolor := < couleur >`, couleur de la grille principale, *gray* par défaut,
- `subgridcolor := < couleur >`, couleur de la grille secondaire, *lightgray* par défaut,
- `gridstyle := < style de ligne >`, indique le style de ligne pour la grille principale, *solid* par défaut,
- `subgridstyle := < style de ligne >`, indique le style de ligne pour la grille secondaire, *solid* par défaut,



Exemple

```
\begin{texgraph}[name=Agrid, file]
Include "grid.mod";
Graph image = [
view(-5,10,-2,5), Marges(0.25,0.25,0.25,0.25), size(12),
draw("grid",[-5,-1+5*i],[nbsubdiv :=[3,3],gridcolor :=maroon]),
SetMatrix([0,1,1+i]),
draw("axes",[0,1+i],Arrows :=1),
draw("grid",[0,5+5*i],[nbsubdiv :=[1,3],subgridstyle :=dashed]),
];
\end{texgraph}
```





FIGURE 5: Exemples de grilles

**Exemple**

```

\begin{texgraph}[name=preface, file]
Include "grid.mod";
Graph image = [
size(8), CurrentArrow := CParrow', FrenchBabel := 1,
draw("axes", [0, pi/2+i],
[grid := 1, limits := [-1.5*pi+1.5*pi*i, jump],
legend := ["$x$", "$y$"], nbsubdiv := [1, 3], labeltext := ["\pi", ""],
labelden := [2, 1], Arrows := 1, gridwidth := 6 ]),
draw("cartesian", sin(x)*3, [x := [-pi, pi], Color := red, Width := 8, legend := "$\mathcal{C}$",
labelpos := 0.75, labeldir := NE,
FillStyle := gradient, GradStyle := radial, FillOpacity := 0.3, FillColorB := darkblue])
];
\end{texgraph}

```




FIGURE 6: Exemple de la page 1